Programming Issues 2





John Findlay  April. 1999





Please don't take offence by anything said herein, nothing mentioned here is meant as a... etc.





Hints and tips herein.





Preamble; the best way to learn is to study other peoples code. If you just try to use their code without understanding it thoroughly, problems will arise that will be difficult to overcome. 

1. FILE HANDLING/ERRORS





It is annoying and unprofessional when a MessageBox appears on the screen saying "GFA BASIC RUNTIME ERROR -> xxx" and then to cap that, the programme terminates.





This should be avoided by trapping errors and dealing with them in an appropriate way. Code should be written with error handling as a priority. This Basic error handling mentioned above is for beginners, and rightly so, but more advanced programmers should develop their techniques and so produce code of a high standard.





Fundamentally there are two ways to deal with errors in your GFA programmes, using the GFA exception handling TRY/CATCH method and explicitly checking for errors yourself.





GFA's TRY/CATCH is a great improvement over the old ON ERROR/RESUME method. You can see an implementation of TRY/CATCH in the LoadGIF listing which you can find on Dales web page. http://www.gfa.net/adbryant/





In this article we will look at the second method, explicit error checking. A personal preference for this method has developed over the years because experience has shown me that 'do-ing it yourself' is the better way. A little more code is required but one's understanding of code-ing is, in my opinion of prime importance if one is to become a proficient programmer.





So, file handling;





One needs to receive a return value for error checking, so use the file handling API's. Below are listed the most commonly used;





API                     Description


---------------------------------------------------------


GetDriveType            Determines the drive type


GetSystemDirectory      Returns Windows system directory


GetTempDrive            Drive letter for temporary files


GetTempFileName         Creates a temporary filename


GetWindowsDirectory     Returns the Windows directory


_hread                  Reads data from a file


_hwrite                 Writes data to a file


_lclose                 Closes an open file


_lcreat                 Creates or opens a file


_llseek                 Repositions the file pointer


_lopen                  Opens an existing file


_lread                  Reads data from a file


_lwrite                 Writes data to a file


OpenFile                Create/open/reopen/delete a file








Check for the return value when opening or creating a file. The fragment of code below shows the type of error handling used in the accompanying listing example2.gfw.





 Define constants in your Init() procedure





  ERR_FILE_ERROR  = -1


  ERR_OPEN_ERROR  = -2








Somewhere in the listing, jump to the function that will open the file;








  zFile$ = zFile$ + chr$(0)


  nRet = @LoadFile(V:zFile$)





  IF hRet > 0


     do something            ' Success


  '==================================


  ' If an error occurs trap the error


  '==================================


  ELSE IF hRet < 0


    @ErrorHandler(nRet)      ' Jump to error handler


  ENDIF








FUNCTION LoadFile(lpzFilename)


  LOCAL hFile





  hFile = _lopen(lpzFilename, OF_READ)





  IF hFile = ERR_FILE_ERROR


    ~_lclose(hFile)


    RETURN ERR_OPEN_ERROR


  ENDIF





  other programme lines





ENDFUNC





PROCEDURE ErrorHandler(nErr)


  LOCAL zError$ = ""





  SELECT nErr





  CASE ERR_FILE_ERROR


    zError$ = "Error reading file!"





  CASE ERR_OPEN_ERROR


    zError$ = "Could not open file!"





  DEFAULT


    zError$ = "Unknown Error! " + STR$(nErr)





  ENDSELECT





  IF zError$ <> ""


    display MessageBox or a custom made dialog


  ENDIF





RETURN








OpenFile() is the more flexible API but for most purposes _lopen() will suffice.








So, now we hopefully have a valid file handle (hFile) let's look at how to trap possible errors when reading and writing to files. In the example2.gfw listing the follow code shows how to check for errors when reading from a file;





  '====================================================


  ' Read the bitmap file header


  '====================================================


  IF LEN(bf.) != _lread(hFile, *bf., LEN(bf.))


    ERASE bf., bc.


    ~_lclose(hFile)


    RETURN ERR_FILE_ERROR


  ENDIF





If the return value does not match the number of bytes you requested to be read an error has occurred so return the error code and deal with it in the PROC ErrorHandler(nErr).





The same applies to writing to a file. Use something like


this;





  nCnt = 1024


  IF nCnt != _lwrite(hFile, lpMem, nCnt)


    ~_lclose(hFile)


    RETURN ERR_FILE_ERROR


  ENDIF





And of course don't forget to close the file!





Ordinarily files are read with no problems but an occasional corrupt file is usually lurking somewhere on everyone's hard drive.











2. MEMORY HANDLING/ERRORS





Another area where errors occur is when allocating memory. The Windows system will usually spill some or the ram's contents out to the swap file and grant your request for a memory block, but for the occasion when this doesn't happen your code should deal with the situation in an elegant manner.





FUNCTION MyFunc(lSize)





  lpMem = MALLOC(lSize)


  IF !lpMem


    RETURN ERR_NO_MEM


  ENDIF





  other programme lines





ENDFUNC








When allocating large memory blocks I suggest that you use the API function GlobalAlloc().





Windows memory management system needs to move memory when necessary, the GFA MALLOC() function allocates fixed (non-movable) memory. With the API function GlobalAlloc() you specify memory flags when requesting the amount of memory you need.





  hbi = GlobalAlloc(GHND, dwSize)





  '====================================================


  ' No memory ?


  '====================================================


  IF !hbi


    RETURN ERR_NO_MEMORY


  ENDIF





The flag GHND combines the GMEM_MOVEABLE and GMEM_ZEROINIT flags. The GMEM_ZEROINIT is convenient as GFA's MALLOC does not initialise memory with zero's.





API GlobalAlloc() returns a handle to the memory block not an address, so before you read or write to the memory you must first lock it with API GlobalLock();





  lphbi = GlobalLock(hbi)





When locked Windows will not move or discard the memory. When you have finished reading or writing to memory you should unlock it with API GlobalUnlock().





  ~GlobalUnlock(hbi)





This may seem long winded if you are used to using MALLOC but if you want you programmes to be professional they have to comply with Windows standards.





This approach to memory management is amply demonstrated in the example2.gfw listing where large memory blocks are needed to store DIB's. I hope you will see that locking and unlocking memory is not such a terrible chore!











3. USING THE SYSTEM COLOURS





Windows is designed so that a user can set the way his/her desktop appears with respect to colours, fonts and sizes of things like scrollbars etc.





You can interrogate the system with a variety of API's to retrieve these parameters. Here, we will look at system colours and how to integrate this into our programmes.





When creating toolbars for example one should not force the user to stare at a button colour that is not of their choosing. Remember they have already decided the colour they want to see for 3D effects by running a programme supplied by MS.





One way to ensure the correct colour for one's toolbar is to store two bitmaps, one for the main image and one for the mask image. By using the API BitBlt() twice with different ROP codes and filling the button with a brush the same colour as the system colour COLOR_BTNFACE one could achieve the desired result.





However there is an easier way,





 use the API CreateMappedBitmap() function.





The CreateMappedBitmap() function loads a bitmap resource and converts one or more colours from the original colour to the desired colour.





CreateMappedBitmap() needs the instance handle of the module that contains the bitmap, the resource id of said bitmap (must be an ordinal number, not a string), a pointer to a colour map struct (or array of colour maps), and the number of colours that you wish to be converted.





The following code shows how this can be done;





Define the colormap structure in your initialise routine and declare the API from the commctrl dll;





  TYPE COLORMAP:


  - LONG from


  - LONG to


  ENDTYPE


  COLORMAP: cm.





  DLL #1,"commctrl"


    DECL WORD CreateMappedBitmap(w,w,w,l,w)


  ENDDLL





And when you want the bitmap use the following code.





Assuming that the original background of the bitmap is grey, change it to COLOR_BTNFACE;





 cm.from = RGB(192, 192, 192)


 cm.to   = GetSysColor(COLOR_BTNFACE)





 hToolBmp = ^CreateMappedBitmap(hInstRes, id, 0, *cm., n)





where;





 hInstRes is the instance handle of the resource module 	





 id       is the ordinal number of the bitmap





 *cm.     is the address of colour maps





 n        is the number of colour maps





You now have a bitmap (hToolBmp) with the right colours.








Another area you need to use COLOR_BTNFACE is when creating GFA dialogs.





Do not use;





	DLG FILL #1, RGB(192, 192, 192)





Use instead;





	DLG FILL #1, GetSysColor(COLOR_BTNFACE)





Here is a list of system colours;





' COLOR_SCROLLBAR           =  0


' COLOR_BACKGROUND          =  1


' COLOR_ACTIVECAPTION       =  2


' COLOR_INACTIVECAPTION     =  3


' COLOR_MENU                =  4


' COLOR_WINDOW              =  5


' COLOR_WINDOWFRAME         =  6


' COLOR_MENUTEXT            =  7


' COLOR_WINDOWTEXT          =  8


' COLOR_CAPTIONTEXT         =  9


' COLOR_ACTIVEBORDER        = 10


' COLOR_INACTIVEBORDER      = 11


' COLOR_APPWORKSPACE        = 12


' COLOR_HIGHLIGHT           = 13


' COLOR_HIGHLIGHTTEXT       = 14


' COLOR_BTNFACE             = 15


' COLOR_BTNSHADOW           = 16


' COLOR_GRAYTEXT            = 17


' COLOR_BTNTEXT             = 18


' COLOR_INACTIVECAPTIONTEXT = 19


' COLOR_BTNHIGHLIGHT        = 20





COLOR_BTNSHADOW, COLOR_BTNFACE and COLOR_BTNHIGHLIGHT are needed if you draw your own 3D effects.





GFA supports these Windows constants up to COLOR_BTNTEXT, if you need to use COLOR_INACTIVECAPTIONTEXT or COLOR_BTNHIGHLIGHT you will have to define them yourself.





COLOR_INACTIVECAPTIONTEXT& = 19


COLOR_BTNHIGHLIGHT&        = 20





Use CreateMappedBitmap() when displaying a bitmap on a suface that has a background colour that will be a system colour.





For dailog's loaded from a resource you only need to add 0x4L to the style of the dialog to have the dialog filled with COLOR_BTNFACE; DS_3DLOOK = 4





STYLE DS_MODALFRAME | 0x4L | WS_POPUP | WS_VISIBLE | 


				WS_CAPTION | WS_SYSMENU





Load the resource, then select 'Edit as text' from the Workshop's menu, type the'0x4L' into the style and save the resource. 











4. RICH TEXT HELP





A simple Rich Text Help is included in the example2.gfw listing. Some of the code and CONSTANTs were taken from Eberhard Funk's RichEdit.GFW programme.





The RTF is stored in the res (dll) using Borlands Resource Workshop.  





This is the method I've used; load into the Workshop either the res file or the resource dll, select 'Add to project' from the file menu and choose 'user resource data' from the drop down selector. Load in your RTF file, you will then be prompted to specify what type of resource. Create a new type, I called it 'RTF', and complete the load. Now save the project as both the res and dll. This allows you to run interpreted and compiled.





Now that the RTF file is in the resource we can construct the code to display it. I have chosen to display it in a dialog box.





We need to use a Windows RichEdit control to display RTF.





I have used CreateWindow() for the RichEdit control with a dialog loaded from the resource, but you could also use the GFA CONTROL command.





Look in the example2.gfw listing in DoCommands() to see where the dialog is created;





  CASE IDM_HELP, IDK_HELP


    IF bComp


      hDlgModal = DialogBox(hInstRes, "helpbox", hWnd, 


			GetProcAddress(NULL, "HelpModal"))


    ELSE


      _CB (8)   = HelpModal(wwwl)


      hDlgModal = DialogBox(hInstRes, "helpbox", hWnd, 


				_CB(8))


    ENDIF





When compiled I use GetProcAddress(), when interpreted I use a GFA _CB. The RTF control is created in the Windows callback procedure HelpModal(hWdlg, wMessage, wParam, lParam).





To load the RTF data into the RichEdit control we need to use another callback. The method used here is to save the resource to disk as a temp file, load that in as a stream, passing the data to Windows to be processed. 





Anyway, here is the code from the example that creates the RTF control and loads the resource. Some lines are split.





The callback that actually loads the data from disk is listed further down; RtfStreamCallback()





  CASE WM_INITDIALOG


    '=========================================


    ' Centre the Dialogbox on the screen


    '=========================================


    ~GetWindowRect(hWdlg, *rect.)


    ~SetWindowPos(hWdlg, NULL, (nSrcW - rect.right) / 2, 


		(nSrcH - rect.bottom) / 2, 0, 0, SWP_NOSIZE)


    ~SetMenu(hWdlg, LoadMenu(hInstRes, "help"))





    '=======================================


    ' Load and lock RTF resource


    '=======================================


    hRtfText = LoadResource(hInstRes, 


			FindResource(hInstRes, "RTF_1", "RTF"))


    lpRtfText = LockResource(hRtfText)





    '==========================================


    ' Save to disk as a temp file


    '==========================================


    hgFile = OpenFile("c:\~emp.dat", *of., OF_CREATE)





    IF hgFile = HFILE_ERROR


      @DoMessageText(NULL, "Could not Initialise Help 


						File!")


      ~SendMessage(hWdlg, WM_USER, nERR_HELP, 0)


    ELSE


      '========================================


      ' You do need to know the size of the resource. 


      ' API's SizeofResource() and GlobalSize() do not 


      ' return the exact size!!!


      '========================================


      ~_lwrite(hgFile, lpRtfText, 9954)


      ~_lclose(hgFile)


      hgFile = 0                    ' Clear hgFile





      '============================================


      ' Open the temp file, create the RichText Window


      '============================================


      hgFile = _lopen("c:\~emp.dat", OF_READ)





      IF hgFile = HFILE_ERROR


        @DoMessageText(hWndMain, "Could not Initialise Help 


							File!")


        ~SendMessage(hWdlg, WM_USER, nERR_HELP, 0)


      ELSE





        hgWndRtf = @CreateRichEditControlForRtf(hWdlg)





        IF !hgWndRtf


          @DoMessageText(hWndMain, "Could not Initialise


						 Help Window!")


          ~SendMessage(hWdlg, WM_USER, nERR_HELP, 0)


        ELSE


          '=======================================


          ' Set up the callback for the RichText Window to 


          ' get data from file


          '=================================


          eStream.dwCookie = hgFile


          eStream.dwError = 0


          IF bComp


            eStream.pfnCallback = GetProcAddress(NULL, 


					"RtfStreamCallback")


          ELSE


            _CB (9)   = RtfStreamCallback(llll)


            eStream.pfnCallback = _CB(9)


          ENDIF





          '============================================


          ' Message EM_STREAMIN starts the callback and 


          ' SF_RTF tells it that the data is Rich Text. 


          '============================================


          ~SendMessage(hgWndRtf, EM_STREAMIN, SF_RTF, 


					*eStream.)





          ~SendMessage(hgWndRtf, EM_SETMODIFY, TRUE, 0)


          ~PostMessage(hWdlg, WM_USER, 0, 0)





        ENDIF


      ENDIF


      '===================================================


      ' Free the resource, close and delete the temp file.


      '===================================================


      ~UnlockResource(hRtfText)


      ~FreeResource(hRtfText)


      ~_lclose(hgFile)


      hgFile = 0               ' Clear hgFile


      ~OpenFile("c:\~emp.dat", *of., OF_DELETE)





    ENDIF


    RETVAL TRUE


    EXPROC








PROCEDURE RtfStreamCallback(dwCookie, lpbBuff, lcb, lpcb)


  $Export RtfStreamCallback





  '============================================


  ' Read the data from the temp file.


  '============================================


  LONG{lpcb} = _lread(dwCookie, lpbBuff, lcb)





  IF BYTE{lpcb} < lcb


    RETVAL FALSE                    ' File is fully read


  ELSE


    RETVAL TRUE                     ' More to read


  ENDIF





RETURN





I suggest you study the code carefully until you have a thorough understanding. Eberhard's RichEdit.GFW has a more fully implemented RichEdit control for you to study.





Another way to include the RFT control is to use Borlands Workshop and add a Custon Control with a "RICHED16" class.





By adding a listbox and a few more RTF resources so that a user can select different pages creates a reasonable help system. You can create your RichText files with WordPad.





I've done one with a listbox and it looks very nice. :-)


 








5. ABOUT DIB's








There are DIB's and there are BITMAP's. DIB's are BITMAP's and BITMAP's can be DIB's, but not always.





Now that's over I will make distinctions for the purposes of this article. DIB's are Device Independent Bitmaps. The bitmaps that we all know and love, that appear as handles after using GFA's LOADBMP() or the API LoadBitmap() are not DIB's they are DDB's. A DDB is a Device Dependant Bitmap.





The difference is important; a DDB or hBmp& if you prefer is always in a format that is the same as the current display. In other words if the image has been loaded whilst the display is set to 32 bits per pixel (bpp), the image will have 32 bpp even if it originated as a 8 bpp (256 colour) image. Conversely if you load a 32 bit DIB whilst the display is set to 8 bpp it will miraculously become an 8 bpp image in Windows memory. The consequence of loading an 8 bpp DIB with LOADBMP() and then saving the resultant hBmp& while the display setting is 32 bpp is horrendous.





This is one reason colours can sometimes be distorted when loading bmp's, because obviously you can't have a 32 bit image displayed properly on an 8 bit display setting. There aren't' enough 'bits'.





All standard Windows bitmaps stored on disk will be DIB's. 





The best way to load, display and save images is as DIB's, this way the original format and size of the file remains intact. The example2 listing loads and displays DIB's. 





There are two new formats introduced with Win95, the 16 bpp and the 32 bpp DIB's. So now the full spectrum is;





BPP     Number of Colours





  1     2 ^ 1    2


  4     2 ^ 2    16


  8     2 ^ 8    256


 16     2 ^ 16   65,536


 24     2 ^ 24   16,777,216


 32     2 ^ 32   4,294,967,296





Actually the 32 bit image only has the same colour resolution as the 24 bit image, we'll see why later.





Also these two new formats (16 & 32 bpp) can include colour masks that informs the Windows DIB machine how to render the images.





The way the image data is arranged for the different formats goes something like this;








1 bpp -> each bit of a byte represent one pixel. It can be on '1' or off '0'. Thus we get a black and white image.





4 bpp -> each nibble (half of a byte) represents one pixel. Four bits (1111) represent 16 different states of on or off, thus we get 16 colours. This format requires a colour palette.





8 bpp -> each pixel is represented by one byte (8 bits). Eight bits can represent 256 colours. This format requires a colour palette.


 


16 bpp -> each pixel is represented by 16 bits. This is a bit tricky because the three colour values of red, green and blue are spread over the 16 bits. The way it's arranged is either as RGB555 or RGB565. 16 bit images can have masks.





The RGB555 pixel data in memory would look like:





 a red pixel   (0111 1100 0000 0000)


 a green pixel (0000 0011 1110 0000)


 a blue pixel  (0000 0000 0001 1111)





The RGB565 pixel data in memory would look like:





 a red pixel   (1111 1000 0000 0000)


 a green pixel (0000 0111 1110 0000)


 a blue pixel  (0000 0000 0001 1111)





Windows is informed of this by the masks that the image stores in the same place as the older type DIB's stored their colour table and the biCompression flag set to BI_BITFIELDS. The value of BI_BITFIELDS is 3.





The RGB555 and RGB565 masks are long words and are defined as;





16bpp RGB555 masks  $00007C00 $000003E0 $0000001F


16bpp RGB565 masks  $0000F800 $000007E0 $0000001F








24 bpp -> each pixel is represented by 24 bits (three bytes). Each byte represents the value of the red, green or blue component of a pixel. The image data for 24 bit images can get quite large.





32 bpp -> each pixel is represented by 24 bits. The other 8 bits are not used. This is why the 32 bpp has the same colour resolution as the 24 bit image. However it is stored as 32 bits so it's even larger then the 24 bit image file.





32 bit images can have masks somewhat like the 16 bit image, but for the 32 bit image the data is in the form of RGB888.





The RGB888 pixel data in memory would look like:





 a red pixel   (0000 0000 1111 1111 0000 0000 0000 0000)


 a green pixel (0000 0000 0000 0000 1111 1111 0000 0000)


 a blue pixel  (0000 0000 0000 0000 0000 0000 1111 1111)








The RGB888 masks are long words and are defined as;





32bpp RGB888 masks  $00FF0000 $0000FF00 $000000FF





The biCompression flag in the DIB header will be one of the following; 








BI_RGB       Specifies that the bitmap is not compressed.


 


BI_RLE8      Specifies a run-length encoded format for 


             bitmaps with 8 bits per pixel. The compression 


             format is a 2-byte format consisting of a 


             count byte followed by a byte containing a 


             colour index. 





BI_RLE4      Specifies a run-length encoded format for 


             bitmaps with 4 bits per pixel. The compression 


             format is a 2-byte format consisting of a 


             count byte followed by two word-length colour 


             indexes.





BI_BITFIELDS Specifies that the bitmap has masks. It will 


             be either a 16 bit or 32 bit image.











The headers and structures used for DIB's are as follows; you will find the headers and a full description for their members at the end of this document for your reference.





First comes the Bitmap File Header





  TYPE BITMAPFILEHEADER:


  - WORD  bfType


  - LONG  bfSize


  - WORD  bfReserved1


  - WORD  bfReserved2


  - LONG  bfOffBits


  ENDTYPE





The BITMAPFILEHEADER structure contains information about the type, size, and layout of a device-independent bitmap (DIB) file. 





A BITMAPINFO or BITMAPCOREINFO structure immediately follows the BITMAPFILEHEADER structure in the DIB file. 





A BITMAPCOREINFO structure was used for Windows 2.0 bitmaps and are rarely seen. These old type bitmaps used RGBTRIPLE structures for storing colour information instead of RGBQUAD's.





Next comes the BITMAPINFO structure which in fact is the BITMAPINFOHEADER and an array of RGBQUAD structures, the number of which depends on how many colours the image palette contains.





If it simpler to ignore this BITMAPINFO structure and just see it as a BITMAPINFOHEADER followed by a number of colour entries.





Here is the BITMAPINFOHEADER;





  TYPE BITMAPINFOHEADER:


  - LONG  biSize


  - LONG  biWidth


  - LONG  biHeight


  - WORD  biPlanes


  - WORD  biBitCount


  - LONG  biCompression


  - LONG  biSizeImage


  - LONG  biXPelsPerMeter


  - LONG  biYPelsPerMeter


  - LONG  biClrUsed


  - LONG  biClrImportant


  ENDTYPE


  BITMAPINFOHEADER: bi.





The BITMAPINFOHEADER structure contains information about the dimensions and colour format of a Windows 3.0 or later device-independent bitmap (DIB). 





And here is the RGBQUAD structure;





  TYPE RGBQUAD:


  - BYTE  rgbBlue


  - BYTE  rgbGreen


  - BYTE  rgbRed


  - BYTE  rgbReserved


  ENDTYPE


  RGBQUAD: rgbq.





They may be up to 256 RGBQUAD entries directly after the BITMAPINFOHEADER depending on how many colours are used by the image. For 16, 24, and 32 bit images there are generally no RGBQUAD entries but there will often be masks for 16 and 32 bit images.











6. LOADING A DIB





Loading a DIB is quite straight forward provided you methodically work through the necessary steps. Please read the following description in conjunction with the listing example2.gfw.





Overview of loading a DIB;





The method for loading a DIB in exampl2.gfw is this;





1.    Open file, if OK jump to ReadDIBBitmapInfo()





2.    Load the file header


         Check and make sure it's a bitmap





3.    Then load first four bytes of bm header


         Check whether old or new type





4.    Then load the bitmap header


         Fill in any missing values





5.    Calculate number of colours





6.    Allocate enough memory to hold DIB header and colours





7.    Load colours or masks if any





8.    Exit routine and return handle to memory block





8.    ReAllocate memory to hold all the pixel bits





9.    Close the file.





Done! Of course the actual code gets a wee bit more complicated but that's the process.





===================================================





Here's some of the code;





  '===========================================


  ' Open the file


  '===========================================


  hFile = _lopen(lpzFilename, OF_READ)





  IF hFile = nFILE_ERROR


    ~_lclose(hFile)


    ERASE bih.


    RETURN nOPEN_ERROR


  ENDIF





After opening the file load the BITMAPFILEHEADER to see if it's a DIB.





  '============================================


  ' Read the bitmap file header


  '============================================


  IF LEN(bf.) != _lread(hFile, *bf., LEN(bf.))


    RETURN nFILE_ERROR


  ENDIF





  '============================================


  ' Do we have a bitmap?


  '============================================


  IF bf.bfType != $4D42 // != CVI("BM")


    RETURN nNOT_VALID


  ENDIF





At this stage we know if it's a bitmap, if it is get the next 4 bytes, this will tell us if it's a BITMAPINFOHEADER or a BITMAPCOREHEADER because the first 4 bytes of both structures contains the size of that structure.





  '================================================


  ' Read the next 4 bytes for header size into dwSize


  '================================================


  IF 4 != _lread(hFile, *dwSize, 4)


    RETURN nFILE_ERROR


  ENDIF





  ~_llseek(hFile, -4, nSEEK_CUR)       ' Go back 4 bytes





Now depending on what type it is we have to either load a BITMAPINFOHEADER or load the BITMAPCOREHEADER. After loading thus far we take a look at various members of the header and calculate the number of colours if any, see what compression is used etc.





IF dwSize is length of BITMAPINFOHEADER  


  load LEN(BITMAPINFOHEADER:)


ELSE


  load LEN(BITMAPCOREHEADER:)


ENDIF





Now we have the header, the various values stored in the header are used to determine what to do.





If it's a BITMAPCOREHEADER we have to convert it to a BITMAPINFOHEADER type so that Windows will be able to display it. This involves filling in some values that are not supplied by the BITMAPCOREHEADER, and after deciding how many colours there are, load the colours.





It header may also have the compression member set to BI_BITFIELDS, if so there will not be any colour entries but there will be masks. These are stored in the same place as the colours would have been.





See the example2.gfw listing for the details.





After the colours/masks and various fields have been sorted we can calculate the size of the image data from the biWidth, biBitCount and biHeight members of the BITMAPINFOHEADER.





In PROCEDURE InitVars() I have defined a function called WIDTHBYTES() that calculates this.





  ' Determines number of BytesPerLine of an image.


  DEFFN WIDTHBYTES(n) = ((((n) + 31) >> 5) << 2)





n = {lpbih}.biWidth * {lpbih}.biBitCount)





Armed with this information we can allocate the memory required to hold the data.





I suggest you study the code carefully until you have a thorough understanding. There are many examples of loading DIB's, mine is only one.





Good luck!





J.M.F April. 1999











7. DIB/BITMAP HEADERS








  TYPE BITMAPFILEHEADER:


  - WORD  bfType


  - LONG  bfSize


  - WORD  bfReserved1


  - WORD  bfReserved2


  - LONG  bfOffBits


  ENDTYPE


  BITMAPFILEHEADER: bf.





The BITMAPFILEHEADER structure contains information about the type, size, and layout of a device-independent bitmap (DIB) file. 





Member	Description


----------------------------------------------------------


bfType	Specifies type of file. This member must be BM. 





bfSize	Specifies the size of the file, in bytes. 





bfReserved1	Reserved, must be set to zero. 





bfReserved2	Reserved, must be set to zero. 





bfOffBits	Specifies the byte offset from the 


		BITMAPFILEHEADER structure to the actual bitmap 


		data in the file. 





==========================================================





  TYPE BITMAPINFOHEADER:


  - LONG  biSize


  - LONG  biWidth


  - LONG  biHeight


  - WORD  biPlanes


  - WORD  biBitCount


  - LONG  biCompression


  - LONG  biSizeImage


  - LONG  biXPelsPerMeter


  - LONG  biYPelsPerMeter


  - LONG  biClrUsed


  - LONG  biClrImportant


  ENDTYPE


  BITMAPINFOHEADER: bi.





The BITMAPINFOHEADER structure contains information about the dimensions and colour format of a Windows 3.0 or later device-independent bitmap (DIB). 





Member	Description


---------------------------------------------------------


biSize       Specifies the number of bytes required by the 


             BITMAPINFOHEADER structure. 





biWidth      Specifies the width of the bitmap, in pixels.


 


biHeight     Specifies the height of the bitmap, in pixels. 





biPlanes     Specifies the number of planes for the target 


		device. This member must be set to 1. 





biBitCount   Specifies the number of bits per pixel. This 


		value must be 1, 4, 8, 16, 24 or 32.


 


biCompression Specifies the type of compression for a 


              bitmap.





It can be one of the following values: 





Value        Meaning


--------------------------------------------------------


BI_RGB       Specifies that the bitmap is not compressed.


 


BI_RLE8      Specifies a run-length encoded format for 


             bitmaps with 8 bits per pixel. The compression 


             format is a 2-byte format consisting of a 


             count byte followed by a byte containing a 


             colour palette index. 





BI_RLE4      Specifies a run-length encoded format for 


             bitmaps with 4 bits per pixel. The compression 


             format is a 2-byte format consisting of a 


             count byte followed by two word-length colour 


             palette indexes.





BI_BITFIELDS Specifies that the bitmap has masks. It will 


             be either a 16 bit or 32 bit image.


---------------------------------------------------------





biSizeImage  Specifies the size, in bytes, of the image. It 


             is valid to set this member to zero if the 


             bitmap is in the BI_RGB format. 





biXPelsPerMeter Specifies the horizontal resolution, in 


                pixels per meter, of the target device for 


                the bitmap. An application can use this 


                value to select a bitmap from a resource 


                group that best matches the characteristics 


                of the current device. 





biYPelsPerMeter Specifies the vertical resolution, in 


                pixels per meter, of the target device for 


                the bitmap. 





biClrUsed      Specifies the number of colour indexes in 


               the colour table actually used by the 


               bitmap. If this value is zero, the bitmap 


               uses the maximum number of colours 


               corresponding to the value of the biBitCount 


               member.








If the biClrUsed member is nonzero, it specifies the actual number of colours that the graphics engine or device driver will access if the biBitCount member is less than 16.





If the bitmap is a packed bitmap (that is, a bitmap in which the bitmap array immediately follows the BITMAPINFO header and which is referenced by a single pointer), the biClrUsed member must be set to zero or to the actual size of the colour table. 





biClrImportant  Specifies the number of colour indexes that 


                are considered important for displaying the 


                bitmap. If this value is zero, all colours 


                are important. 





==========================================================





  TYPE BITMAPCOREHEADER:


  - LONG  bcSize


  - WORD  bcWidth


  - WORD  bcHeight


  - WORD  bcPlanes


  - WORD  bcBitCount


  ENDTYPE





Member	Description


---------------------------------------------------------


bcSize       Specifies the number of bytes required by the 


             BITMAPCOREHEADER structure. 





bcWidth      Specifies the width of the bitmap, in pixels. 





bcHeight     Specifies the height of the bitmap, in pixels. 





bcPlanes     Specifies the number of planes for the target 


             device. This member must be set to 1. 





bcBitCount  Specifies the number of bits per pixel. This 


            value must be 1, 4, 8, or 24. 





=========================================================





  TYPE RGBQUAD:


  - BYTE  rgbBlue


  - BYTE  rgbGreen


  - BYTE  rgbRed


  - BYTE  rgbReserved


  ENDTYPE





  TYPE RGBTRIPLE:


  - BYTE  rgbtBlue


  - BYTE  rgbtGreen


  - BYTE  rgbtRed


  ENDTYPE














